

Challenge September 2024

Boat Rental

A solution with OPL CPLEX by Alex Fleischer
afleischer@fr.ibm.com

CPLEX is available within IBM AI platform watsonx.AI
And OPL is a high level modeling language for CPLEX

We can write

```
tuple boat
{
  key string name;
  string manufacturer;
  float cost;
  int seats;
  float profit;
}

{string} manufacturers={"Sleekboat","Racer"};

{boat} boats with manufacturer in manufacturers=
{
  <"Speedhawk","Sleekboat",6000,3,70>,
  <"Silverbird","Sleekboat",7000,5,80>,
  <"Catman","Racer",5000,2,50>,
  <"Classy","Racer",9000,6,110>
};
```

```

// how many boats shall we buy ?
dvar int+ buy[boats];

dexpr float profit=sum(b in boats) buy[b]*b.profit;;

maximize profit;

subject to
{
//Floataway tours has $420,000
sum(b in boats) buy[b]*b.cost<=420000;

// Floataway tours would like to purchase at least 50 boats
sum(b in boats) buy[b]>=50;

// purchase the same number from Sleekboat as from Racer to
maintain goodwill
forall(ordered i,j in manufacturers)
    sum(<x,i,y,z,t> in boats) buy[<x,i,y,z,t>]==
    sum(<x,j,y,z,t> in boats) buy[<x,j,y,z,t>];

// have a total capacity of at least 200.
sum(b in boats) buy[b]*b.seats>=200;
}
execute display
{
    writeln("profit = ",profit);
    for(var b in boats) if (buy[b]!=0) writeln(buy[b]," * ",b.name);
}

/*
which gives

```

```

profit = 5040
28 * Speedhawk
28 * Classy
*/

```